Advanced Topics from Scientific Computing

TU Berlin Winter 2023/24

Notebook 18

 Jürgen Fuhrmann

CairoMakie

```
begin
    using Test
    using VoronoiFVM,ExtendableGrids
    using OrdinaryDiffEq
    using LinearAlgebra
    using PlutoUI, HypertextLiteral,UUIDs
    using DataStructures
    using GridVisualize
    import CairoMakie
    CairoMakie.activate!(type="svg")
    default_plotter!(CairoMakie)
end
```

Table of Contents


```
TableOfContents()
```

1D Nonlinear Diffusion

Solve the nonlinear diffusion equation

$$\partial_t u - \Delta u^m = 0$$

in $\Omega = (-1, 1)$ with homogeneous Neumann boundary conditions.

This equation is also called "porous medium equation".

For space dimension d in the domain $\mathbb{R}^d \times (0, \infty)$ the equation has a radially symmetrict exact solution, the so-called Barenblatt solution:

$$b(x,t) = \max\left(0, t^{-\alpha}\left(1 - \frac{\alpha(m-1)|x|^2}{2dmt^{\frac{2\alpha}{d}}}\right)^{\frac{1}{m-1}}\right)$$

Here, $\alpha = \frac{1}{m-1+\frac{2}{d}}$. (see Barenblatt, G. I. "On nonsteady motions of gas and fluid in porous medium." Appl. Math. and Mech.(PMM) 16.1 (1952): 67-78.)

We initialize this problem with the exact solution for $t = t_0 = 0.001$.

```
function barenblatt(x,t,m)
    t1=t^(-1.0/(m+1.0))
    x1=1- (x*t1)^2*(m-1)/(2.0*m*(m+1))
    x1<0.0 ? 0.0 : t1*x1^(1.0/(m-1.0))
end;
```

```
0.01
```

```
begin
    const m=2
    const ε=1.0e-10
    const n=50
    const t0=1.0e-3
    const tend=1.0e-2
end
```

```
X = -1.0:0.04:1.0
```

```
X=range(-1,1,length=n+1)
```

```
grid = ExtendableGrids.ExtendableGrid{Float64, Int32};
       dim: 1 nodes: 51 cells: 50 bfaces: 2
```

```
grid=simplexgrid(X)
```

```
nld_flux! (generic function with 1 method)
```

```
    function nld_flux!(f, u, edge)
        f[1] = u[1, 1]^m - u[1, 2]^m
    end
```

```
nld_storage! (generic function with 1 method)
```

```
function nld_storage!(f, u, node)
    f[1] = u[1]
end
```

```
nld_fvm_sys =
VoronoiFVM.System{Float64, Float64, Int32, Int64, Matrix{Int32}, Matrix{Float64}}(num_spe
```

```
nld_fvm_sys = VoronoiFVM.System(grid, storage=nld_storage!, flux=nld_flux!,
species=[1])
```

```
view(::Matrix{Float64}, 1, :): [0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.
```

```
begin
    nld_inival = unknowns(nld_fvm_sys)
    nld_inival[1,:].=barenblatt.(X,t0,m)
end
```




```
scalarplot(grid,nld_inival[1,:], size=(600,200))
```

```
nld_fvm_sol =
t: 109-element Vector{Float64}:
 0.001
 0.0010001
 0.00100022
 0.0010003640000000001
 0.0010005368
 0.00100074416
 0.001000992992
 ⋮
 0.008695060666880321
 0.009021295500160242
 0.0093475303344016
 0.009565020222293441
 0.009782510111146722
 0.01
u: 109-element Vector{Matrix{Float64}}:
 [0.0 0.0 … 0.0 0.0]
 [0.0 0.0 … 0.0 0.0]
 [0.0 0.0 … 0.0 0.0]
 [0.0 0.0 … 0.0 0.0]
 [0.0 0.0 … 0.0 0.0]
 [0.0 0.0 … 0.0 0.0]
 [0.0 0.0 … 0.0 0.0]
 ⋮
 [9.286214107895018e-227 2.7246035877781935e-111 … 2.724603587778198e-111 9.2862141078950
 [2.6673091661702735e-193 1.6739667796753444e-95 … 1.6739667796753448e-95 2.6673091661702
 [3.469942273025075e-166 2.303046131847799e-82 … 2.303046131847811e-82 3.46994227302509e-
 [1.40875416816348e-155 7.722602656643463e-78 … 7.722602656643491e-78 1.4087541681634915e
 [1.1700092284694368e-143 7.268450326335965e-72 … 7.268450326335984e-72 1.170009228469442
 [5.429642603991761e-132 4.816385223877478e-66 … 4.816385223877491e-66 5.4296426039917934
```

```
1 nld_fvm_sol=solve(nld_fvm_sys;inival=nld_inival,times=(t0,tend), Δt=1.0e-7,
  Δt_min=1.0e-7, log=true)
```

Implementation with ODE solvers from DifferentialEquations.jl

```
diffeqmethods =
  OrderedDict("QNDF2 (Like matlab's ode15s)" ⇒ QNDF2, "Rodas5" ⇒ Rodas5, "Rosenbrock23 (
```

method: QNDF2 (Like matlab's ode15s)

```
t: 36-element Vector{Float64}:
 0.001
 0.001001
 0.0010019999999999999
 0.0010119999999999999
 0.0010849458833568828
 0.0011578917667137657
 0.0012713301128828227
 ⋮
 0.00748386371135225
 0.00794330792479726 7
 0.008451501159931527
 0.009009991776161599
 0.009568482392391671
 0.01
u: 36-element Vector{Matrix{Float64}}:
 [0.0 0.0 … 0.0 0.0]
 [0.0 0.0 … 0.0 0.0]
 [0.0 0.0 … 0.0 0.0]
 [0.0 0.0 … 0.0 0.0]
 [0.0 0.0 … 0.0 0.0]
 [0.0 0.0 … 0.0 0.0]
 [0.0 0.0 … 0.0 0.0]
 ⋮
 [0.0 1.9257524929339817e-229 … 1.9257524929339978e-229 0.0]
 [0.0 3.3739495435492346e-213 … 3.373949543549274e-213 0.0]
 [0.0 3.7268906591261018e-174 … 3.726890659126096e-174 0.0]
 [5.399550227693583e-267 3.718280693385946e-126 … 3.7182806933859614e-126 5.3995502276936
 [4.108628451736171e-182 8.853380370353453e-84 … 8.853380370353564e-84 4.1086284517361896
 [9.710681826587577e-161 6.195363443603359e-78 … 6.195363443603411e-78 9.710681826587682e
```

```
begin
    nld_ode_sys=VoronoiFVM.System(grid, storage=nld_storage!, flux=nld_flux!,
    species=[1])
    nld_problem = ODEProblem(nld_ode_sys,nld_inival,(t0,tend))
    odesol=solve(nld_problem,
        diffeqmethods[nld_method](),
        adaptive=true,
         reltol=1.0e-3,
        abstol=1.0e-3,
        initializealg=NoInit()
        )
        nld_ode_sol=reshape(odesol,nld_ode_sys)
end
```



0.001

```
@bind nld_time Slider(range(t0,tend,length=101),show_value=true)
```

1D Nonlinear Storage

This equation comes from the transformation of the nonlinear diffusion equation

$$\partial_t v - \Delta v^m = 0$$

to

$$\partial_t u^{\frac{1}{m}} - \Delta u = 0$$

in $\Omega = (-1, 1)$ with homogeneous Neumann boundary conditions. We can derive an exact solution from the Barenblatt solution of the equation for u.

```
u0 =
```

```
[0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 2.15
```

```
u0=map(x->barenblatt(x,t0,m)^m,X)
```

Direct implementation with VoronoiFVM

```
nls_flux! (generic function with 1 method)
```

```
    function nls_flux!(f,u,edge)
        f[1]=u[1,1]-u[1,2]
    end
```

Storage term needs to be regularized as its derivative at 0 is infinity:

```
nls_storage! (generic function with 1 method)
```

```
function nls_storage!(f,u,node)
    f[1]=(ε+u[1])^(1.0/m)
end
```

```
(seconds = 2.79, tasm = 0.14, tlinsolve = 0.0162, steps = 731, iters = 2920, maxabsnorm
```

```
begin
    nls_sys=VoronoiFVM.System(grid;flux=nls_flux!, storage=nls_storage!,species=[1])
    nls_inival=unknowns(nls_sys)
    nls_inival[1,:].=u0
    nls_sol=VoronoiFVM.solve(nls_sys;inival=nls_inival,times=(t0,tend),Δt_min=1.0e-4,Δt=1.0e-4,Δu_opt=0.1,force_first_step=true,log=true)
    history_summary(nls_sol)
end
```

Implementation as DAE

If we want to solve the problem with standard ODE solvers, we see that the problem structure does not fit into the setting of that package due to the nonlinearity under the time derivative. Here we propose a reformulation to a DAE as a way to achieve this possibility:

$$\begin{cases} \partial_t w - \Delta u & = 0 \\ w^m - u & = 0 \end{cases}$$

```
dae_storage! (generic function with 1 method)
```

```
function dae_storage!(y,u,node)
    y[1]=u[2]
end
```

```
dae_reaction! (generic function with 1 method)
```

```
function dae_reaction!(y,u,node)
    y[2]= u[2]^m-u[1]
end
```

First, we test this with the implicit Euler method of VoronoiFVM

```
(seconds = 2.35, tasm = 0.122, tlinsolve = 0.0172, steps = 732, iters = 2205, maxabsnorm
```

```
begin
    dae_sys=VoronoiFVM.System(grid;flux=nls_flux!, storage=dae_storage!,
    reaction=dae_reaction!,species=[1,2])
    dae_inival=unknowns(dae_sys)
    dae_inival[1,:].=u0
    dae_inival[2,:].=u0.^(1/m)
    dae_control=VoronoiFVM.SolverControl()
    dae_sol=VoronoiFVM.solve(dae_sys;inival=dae_inival,times=(t0,tend),Δt_min=1.0e-
4,Δt=1.0e-4,Δu_opt=0.1,force_first_step=true,log=true)
    history_summary(dae_sol)
end
```

Implementation via OrdinaryDiffEq.jl

method: QNDF2 (Like matlab's ode15s)

```
begin
    dae_ode_sys=VoronoiFVM.System(grid;flux=nls_flux!, storage=dae_storage!,
    reaction=dae_reaction!,species=[1,2])

    dae_ode_problem = ODEProblem(dae_ode_sys,dae_inival,(t0,tend))
    odesol2=solve(dae_ode_problem,
        diffeqmethods[method](),
        adaptive=true,
         reltol=1.0e-3,
        abstol=1.0e-3,
        initializealg=NoInit()
        )
        dae_ode_sol=reshape(odesol2,dae_ode_sys)
end;
```



t= 0.0019999

```
plotsolutions (generic function with 1 method)
```

```
function plotsolutions(t)
    vis=GridVisualizer(resolution=(600,300),dim=1,Plotter=CairoMakie,legend=:lt);
    u=nls_sol(t)
    u_dae=dae_sol(t)
    u_de=dae_ode_sol(t)
    scalarplot!(vis,X,map(x->barenblatt(x,t,m).^m,X),clear=true,color=:red,linestyle=:solid,flimits=(0,100),label="exact")
    scalarplot!(vis,grid,u_dae[1,:],clear=false,color=:green,linestyle=:solid,label="vfvm_dae")
    scalarplot!(vis,grid,u_de[1,:],clear=false,color=:blue,markershape=:cross,linestyle=:dot,label="vfvm_diffeq")
    scalarplot!(vis,grid,u[1,:],clear=false,color=:black,markershape=:none,linestyle=:dash,title="t=$(t)",label="vfvm_default")
    reveal(vis)
end
```

Brusselator system

Two species diffusing and interacting via a reaction

$$\partial_t u_1 - \nabla \cdot (D_1 \nabla u_1) + (B+1)u_1 - A - u_1^2 u_2 = 0$$
$$\partial_t u_2 - \nabla \cdot (D_2 \nabla u_2) + u_1^2 u_2 - B u_1 = 0$$

```
begin
    const bruss_A=2.25
    const bruss_B=7.0
    const bruss_D_1=0.025
    const bruss_D_2=0.25
    const pert=0.1
    const bruss_tend=150
end;

```

```
function bruss_storage(f,u,node)
    f[1]=u[1]
    f[2]=u[2]
end;

```

```
function bruss_diffusion(f,u,edge)
    f[1]=bruss_D_1*(u[1,1]-u[1,2])
    f[2]=bruss_D_2*(u[2,1]-u[2,2])
end;

```

```
function bruss_reaction(f,u,node)
    f[1]= (bruss_B+1.0)*u[1]-bruss_A-u[1]^2*u[2]
    f[2]= u[1]^2*u[2]-bruss_B*u[1]
end;

```

```
begin
    bruss_X=-1:0.1:1
    bruss_grid=simplexgrid(bruss_X,bruss_X)
    bruss_system=VoronoiFVM.System(bruss_grid,species=[1,2],
            flux=bruss_diffusion, storage=bruss_storage, reaction=bruss_reaction);
    bruss_inival=unknowns(bruss_system,inival=0)
    coord=bruss_grid[Coordinates]
    fpeak(x)=exp(-norm(10*x)^2)
    for i=1:size(bruss_inival,2)
            bruss_inival[1,i]=fpeak(coord[:,i])
            bruss_inival[2,i]=0
    end

end
```

```
begin
bruss_ode_problem = ODEProblem(bruss_system,bruss_inival,(0,bruss_tend))
    odesol3=solve(bruss_ode_problem,
        diffeqmethods[bruss_method](),
        adaptive=true,
        reltol=1.0e-3,
        abstol=1.0e-3,
        initializealg=NoInit()
        )
bruss_tsol=reshape(odesol3,bruss_system)
end;
```

method: QNDF2 (Like matlab's ode15s)

t: 150.0




```
let
bruss_sol=bruss_tsol(t_bruss);

vis=GridVisualizer(;layout=(1,2),size=(600,300))
    scalarplot!(vis[1,1],bruss_grid,bruss_sol[1,:],limits=
    (0,10),show=true,colormap=:summer)
    scalarplot!(vis[1,2],bruss_grid,bruss_sol[2,:],limits=
    (0.5,3),show=true,colormap=:summer)
end

```




```
scalarplot(bruss_tsol.t[1:end-1],bruss_tsol.t[2:end]-bruss_tsol.t[1:end-
1],yscale=:log,resolution=(500,200),xlabel="t",ylabel="Δt",title="timesteps")
```